NAG Fortran Library Routine Document

F07BFF (DGBEQU)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07BFF (DGBEQU) computes diagonal scaling matrices D_R and D_C intended to equilibrate a real m by n band matrix A of band width $(k_l + k_u + 1)$, and reduce its condition number.

2 Specification

```
SUBROUTINE F07BFF (M, N, KL, KU, AB, LDAB, R, C, ROWCND, COLCND, AMAX,
1                  INFO)
INTEGER            M, N, KL, KU, LDAB, INFO
double precision   AB(LDAB,*), R(*), C(*), ROWCND, COLCND, AMAX
```

The routine may be called by its LAPACK name ***dgbequ***.

3 Description

F07BFF (DGBEQU) computes the diagonal scaling matrices. The diagonal scaling matrices are chosen to try to make the elements of largest absolute value in each row and column of the matrix B given by

$$B = D_R A D_C$$

have absolute value 1. The diagonal elements of D_R and D_C are restricted to lie in the safe range $(\delta, 1/\delta)$, where δ is the value returned by routine X02AMF. Use of these scaling factors is not guaranteed to reduce the condition number of A but works well in practice.

4 References

None.

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: M ≥ 0.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: N ≥ 0.

3: KL – INTEGER *Input*

On entry: k_l, the number of subdiagonals of the matrix A.

Constraint: KL ≥ 0.

4: KU – INTEGER *Input*

On entry: k_u, the number of superdiagonals of the matrix A.

Constraint: KU ≥ 0.

5: AB(LDAB,$*$) – ***double precision*** array *Input*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the band matrix A whose scaling factors are to be computed, stored in rows 1 to $k_l + k_u + 1$ of AB. The jth column of the matrix A is stored in the jth column of the array AB as follows:

$$\mathrm{AB}(k_u + 1 + i - j, j) = a_{ij}, \quad \text{for } \max(1, j - k_u) \leq i \leq \min(m, j + k_l).$$

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07BFF (DGBEQU) is called.

Constraint: $\mathrm{LDAB} \geq \mathrm{KL} + \mathrm{KU} + 1$.

7: R($*$) – ***double precision*** array *Output*

Note: the dimension of the array R must be at least $\max(1, \mathrm{M})$.

On exit: if INFO $= 0$ or INFO $>$ M, R contains the row scale factors, the diagonal elements of D_R. The elements of R will be positive.

8: C($*$) – ***double precision*** array *Output*

Note: the dimension of the array C must be at least $\max(1, \mathrm{N})$.

On exit: if INFO $= 0$, C contains the column scale factors, the diagonal elements of D_C. The elements of C will be positive.

9: ROWCND – ***double precision*** *Output*

On exit: if INFO $= 0$ or INFO $>$ M, ROWCND contains the ratio of the smallest value of R(i) to the largest value of R(i). If ROWCND ≥ 0.1 and AMAX is neither too large nor too small, it is not worth scaling by D_R.

10: COLCND – ***double precision*** *Output*

On exit: if INFO $= 0$, COLCND contains the ratio of the smallest value of C(i) to the largest value of C(i).

If COLCND ≥ 0.1, it is not worth scaling by D_C.

11: AMAX – ***double precision*** *Output*

On exit: $\max |a_{ij}|$. If AMAX is very close to overflow or underflow, the matrix A should be scaled.

12: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0 and INFO $\leq$ M

If INFO $= i$, the ith row of A is exactly zero.

INFO $>$ M

If INFO $= i$, the $(i - \mathrm{M})$th column of A is exactly zero

7 Accuracy

The computed scale factors will be close to the exact scale factors.

8 Further Comments

The complex analogue of this routine is F07BTF (ZGBEQU).

9 Example

To equilibrate the band matrix A given by

$$A = \begin{pmatrix} -0.23 & 2.54 & -3.66 \times 10^{-10} & 0 \\ -6.98 \times 10^{10} & 2.46 \times 10^{10} & -2.73 & -2.13 \times 10^{10} \\ 0 & 2.56 & 2.46 \times 10^{-10} & 4.07 \\ 0 & 0 & -4.78 \times 10^{-10} & -3.82 \end{pmatrix}.$$

Details of the scaling factors, and the scaled matrix are output.

9.1 Program Text

```
*     F07BFF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KLMAX, KUMAX
      PARAMETER        (NMAX=8,KLMAX=4,KUMAX=4)
      INTEGER          LDAB
      PARAMETER        (LDAB=KLMAX+KUMAX+1)
*     .. Local Scalars ..
      DOUBLE PRECISION AMAX, BIG, CJ, COLCND, ROWCND, SMALL
      INTEGER          I, IFAIL, INFO, J, K, KL, KU, N
*     .. Local Arrays ..
      DOUBLE PRECISION AB(LDAB,NMAX), C(NMAX), R(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION X02AJF, X02AMF
      INTEGER          X02BHF
      EXTERNAL         X02AJF, X02AMF, X02BHF
*     .. External Subroutines ..
      EXTERNAL         DGBEQU, X04CEF
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07BFF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KL, KU
      IF (N.LE.NMAX .AND. KL.LE.KLMAX .AND. KU.LE.KUMAX) THEN
*
*        Read the band matrix A from data file
*
         K = KU + 1
         READ (NIN,*) ((AB(K+I-J,J),J=MAX(I-KL,1),MIN(I+KU,N)),I=1,N)
*
*        Print the matrix A
*
         IFAIL = 0
         CALL X04CEF(N,N,KL,KU,AB,LDAB,'Matrix A',IFAIL)
         WRITE (NOUT,*)
*
```

```
*        Compute row and column scaling factors
*
         CALL DGBEQU(N,N,KL,KU,AB,LDAB,R,C,ROWCND,COLCND,AMAX,INFO)
*
         IF (INFO.GT.0) THEN
            IF (INFO.LE.N) THEN
               WRITE (NOUT,99999) 'Row ', INFO, ' of A is exactly zero'
            ELSE
               WRITE (NOUT,99999) 'Column ', INFO - N,
     +           ' of A is exactly zero'
            END IF
         ELSE
*
*           Print ROWCND, COLCND, AMAX and the scale factors
*
            WRITE (NOUT,99998) 'ROWCND = ', ROWCND, ', COLCND = ',
     +        COLCND, ', AMAX = ', AMAX
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Row scale factors'
            WRITE (NOUT,99997) (R(I),I=1,N)
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Column scale factors'
            WRITE (NOUT,99997) (C(I),I=1,N)
            WRITE (NOUT,*)
*
*           Compute values close to underflow and overflow
*
            SMALL = X02AMF()/(X02AJF()*X02BHF())
            BIG = 1.0D0/SMALL
            IF ((ROWCND.GE.0.1D0) .AND. (AMAX.GE.SMALL)
     +          .AND. (AMAX.LE.BIG)) THEN
               IF (COLCND.LT.0.1D0) THEN
*
*                 Just column scale A
*
                  DO 40 J = 1, N
                     CJ = C(J)
                     K = KU + 1 - J
                     DO 20 I = MAX(1,J-KU), MIN(N,J+KL)
                        AB(K+I,J) = AB(K+I,J)*CJ
   20                CONTINUE
   40             CONTINUE
*
*                 Print the column scaled matrix
*
                  IFAIL = 0
                  CALL X04CEF(N,N,KL,KU,AB,LDAB,'Scaled matrix',IFAIL)
*
               END IF
            ELSE IF (COLCND.GE.0.1D0) THEN
*
*              Just row scale A
*
               DO 80 J = 1, N
                  K = KU + 1 - J
                  DO 60 I = MAX(1,J-KU), MIN(N,J+KL)
                     AB(K+I,J) = R(I)*AB(K+I,J)
   60             CONTINUE
   80          CONTINUE
*
*              Print the row scaled matrix
*
               IFAIL = 0
               CALL X04CEF(N,N,KL,KU,AB,LDAB,'Scaled matrix',IFAIL)
*
            ELSE
*
*              Row and column scale A
*
               DO 120 J = 1, N
                  CJ = C(J)
```

```
                  K = KU + 1 - J
                  DO 100 I = MAX(1,J-KU), MIN(N,J+KL)
                     AB(K+I,J) = R(I)*AB(K+I,J)*CJ
  100             CONTINUE
  120          CONTINUE
*
*              Print the row and column scaled matrix
*
               IFAIL = 0
               CALL X04CEF(N,N,KL,KU,AB,LDAB,'Scaled matrix',IFAIL)
*
            END IF
         END IF
      ELSE
         WRITE (NOUT,*)
     +     'One or more of NMAX, KLMAX or KUMAX is too small'
      END IF
      STOP
*
99999 FORMAT (1X,A,I4,A)
99998 FORMAT (1X,3(A,1P,E7.1))
99997 FORMAT ((1X,1P,7E11.1))
      END
```

9.2 Program Data

```
F07BFF Example Program Data
  4  1  2                                        :Values of N, KL and KU
 -2.30D-01   2.54D+00  -3.66D-10
 -6.98D+10   2.46D+10  -2.73D+00  -2.13D+10
             2.56D+00   2.46D-10   4.07D+00
                       -4.78D-10  -3.82D+00   :End of matrix A
```

9.3 Program Results

```
 F07BFF Example Program Results

 Matrix A
                1            2            3            4
 1   -2.3000E-01   2.5400E+00  -3.6600E-10
 2   -6.9800E+10   2.4600E+10  -2.7300E+00  -2.1300E+10
 3                 2.5600E+00   2.4600E-10   4.0700E+00
 4                             -4.7800E-10  -3.8200E+00

 ROWCND = 3.6E-11, COLCND = 1.4E-10, AMAX = 7.0E+10

 Row scale factors
     3.9E-01    1.4E-11    2.5E-01    2.6E-01

 Column scale factors
     1.0E+00    1.0E+00    6.9E+09    1.0E+00

 Scaled matrix
          1       2       3       4
 1  -0.0906  1.0000 -1.0000
 2  -1.0000  0.3524 -0.2714 -0.3052
 3           0.6290  0.4195  1.0000
 4                  -0.8684 -1.0000
```